Newsgroups: bit.listserv.cmspip-l
Date:         Wed, 16 Dec 1992 10:49:32 EST
From:         Wayne Smith <WTS@MAINE.MAINE.EDU>
Subject:      Re: Screen capture

Chris 'Dero' DeRoberts writes ...
>Does anyone know of, or have, a PIPEs, or PIPEs + REXX + whatever,
>screen capture prgram for VM? Basically, I'm looking for something
>that will intercept the PRINT SCREEN key on various IBM consoles and
>redirect the output to disk and/or a spool file (via PIPE's "PRINTMC").

   In the "whatever" category is the MPVM facility of PVM V2 (VM
   Passthrough).  MPVM is a session manager of sorts (you need to logon
   to CMS first).  You can then start multiple sessions.  Then, in a
   number of ways you can invoke a REXX Macro.  The MPVM subcom
   interface allows you wonderful control of exactly what happens next.

   For example, I recently wrote a NAMES file entry and MPVM macro to
   automatically DIAL VTAM, select from the VTAM screen, and pre-fill our
   CICS signon panel.  PF24 in this example does a print-screen to the
   spooled CMS printer.  Total time to do this?  a few hours, but that
   was because I knew nothing about MPVM facilities when I started.

   Though most of the comments are specifically directed to folks on our
   Maine system, I reproduce SCRN_PRT MPVM below (at least it has a pipe)

/* SCRN_PRT MPVM - my 1st MPVM macro .. have a heart and don't laugh :-)

   Dump the current MPVM screen to the spooled printer of the
   host CMS session.

   No parameters are expected nor processed.

   Note: This MPVM exec might be called via a PF key in an MPVM
         session.

   Example 1: Create an MPVM NAMES file and add the following entry
         (this entry is called "TEST" and may be used for a session on
         the MAINE node.  It will be invoked by the CMS command
         "MPVM * TEST".  Change the names TEST and MAINE to suit your
         purposes.):

           :nick.TEST
              :target.MAINE :PF24.MACRO SCRN_PRT

         (Optionally add (change "WTS" to your id)):
              :?.COMMAND     SET VIEW OFF
              :signon.MACRO  INTERP WAIT 1 500;TAB;TAB;ENTER DIAL VTAM
              :initcmd.MACRO INTERP ENTER C2;WAIT 0 250;
                             PLACE "/WTS/";TAB;
                             SET VIEW ON;REFRESH
   History:

     11/92 original version  WTS@MAINE.MAINE.EDU Wayne Smith, CAPS
     12/92 add auto-connect  to VTAM & VSEPROD (C2)   wts@maine.maine.edu
*/

trace
'COMMAND EXTRACT /DISPLAY/'     /* extract contents of screen */
if rc<>0 then do
   say 'Error code' rc 'in SCRN_PRT MPVM.'
   exit rc
   end
'PIPE (end ?)',
   '       stem mpvm_display.',              /* the whole screen */
   '|rest: take 1',                          /* skip to new page */
   '|      spec "1" 1.1 1-* 2',              /*    ... add cc    */
   '|all:  faninany',                 /* merge 1st line & others */
   '|      asatomc',                         /* ...              */
   '|      printmc',                         /*     and print    */
   '?rest:',
   '|      spec " " 1.1 1-* 2',      /* single space other lines */
   '|all:'
retc = rc
'PIPE cp CLOSE PRT'
exit retc
Comments: Gated by NETNEWS@AUVM.AMERICAN.EDU